Utilisation de *Maxima* pour la résolution du sujet 25 de l'épreuve expérimentale en Terminale S

10 mai 2007

Vous savez sans doute qu'il y aura l'an prochain une épreuve de travaux pratiques au baccalauréat série S.
Des sujets expérimentaux ont été testés dans plusieurs lycées et je vous propose de regarder le sujet 25 de la liste.

Soit (v) la suite définie par : $u_n = \frac{1}{n}\sum_{k=1}^{k=n} k(k-1)$ pour tout n entier naturel non nul.

Déclarons la suite :

```
> u[n]:=(1/n)*sum(k*(k-1), k, 1, n);
```

Calculons et représentons graphiquement les seize premiers termes de cette suite :
Représentons graphiquement les résultats obtenus :

```
> xx:makelist(n, n, 1, 16);
```

$$[1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16]$$

```
> yy:makelist(u[n], n, 1, 16);
```

$$\left[0,1,\frac{8}{3},5,8,\frac{35}{3},16,21,\frac{80}{3},33,40,\frac{143}{3},56,65,\frac{224}{3},85\right]$$

```
> plot2d([discrete, xx, float(yy)]);
```



Définissons maintenant la suite (v) de la manière suivante : pour tout $n \in \mathbb{N}^*$, $v_n = 3u_n$.
Calculons et représentons graphiquement les seize premiers termes de cette suite :
Représentons graphiquement les résultats obtenus :

```
> v[n]:=3*u[n]$
```

```
> zz:makelist(v[n], n, 1, 16);
```

$$[0,3,8,15,24,35,48,63,80,99,120,143,168,195,224,255]$$

```
> plot2d([discrete, xx, zz]);
```



Nous remarquons, au regard du graphique obtenu, qu'il doit y avoir une fonction polynôme du second degré derrière cela.
La suite (v) serait-elle définie par : $v_n = f(n)$ où f serait une fonction polynomiale du second degré ?

```
> f(n):=a*n^2+b*n+c;
```

Essayons de trouver a, b et c en résolvant un système linéaire obtenu en égalisant $f(n)$ à v_n pour trois valeurs de n.

```
> linsolve([v[1]=f(1), v[2]=f(2), v[3]=f(3)], [a, b, c]);
```

$$[a=1, b=0, c=-1]$$

Déterminons les seize premières valeurs de $f(n)$ et de v_n.

```
> f(n):=n^2-1;
```

```
> makelist(f(n), n, 1, 16);
```

$$[0,3,8,15,24,35,48,63,80,99,120,143,168,195,224,255]$$

```
> makelist(v[n], n, 1, 16);
```

$$[0,3,8,15,24,35,48,63,80,99,120,143,168,195,224,255]$$

Nous pouvons conjecturer que $v_n = f(n) = n^2 - 1$. Ceci nous permet de supposer que $u_n = \dfrac{1}{3}(n^2-1)$, pour tout $n \in \mathbb{N}^*$. Reste à démontrer ce résultat rigoureusement, par récurrence pourquoi pas.